Corrigé du TP Maple n°5

Exercice 1 : mouvement de roulis d'un bateau

```
> restart:
> deq:=diff(theta(t),t$2)+k*diff(theta(t),t)+theta(t)+theta
  (t)^2=a*sin(omega*t);
```

$$deq := \frac{d^2}{dt^2}\,\theta(t) + k\left(\frac{d}{dt}\,\theta(t)\right) + \theta(t) + \theta(t)^2 = a\sin(\omega t)$$

```
> s:=dsolve(deq,theta(t));
```

$$s :=$$

Pas de solution. On résout numériquement :

```
> k:=0.1;omega:=0.85;a:=0.0752;
```

$$k := 0.1$$

$$\omega := 0.85$$

$$a := 0.0752$$

```
> init:=theta(0)=0,D(theta)(0)=0;
```

$$init := \theta(0) = 0,\ D(\theta)(0) = 0$$

```
> s:=dsolve({deq,init},theta(t),numeric);
```

$s :=$ **proc**(x_rkf45) ... **end proc**

```
> x:=t->rhs(s(t)[2]);y:=t->rhs(s(t)[3]);
```

$$x := t \rightarrow rhs\left(s(t)_2\right)$$

$$y := t \rightarrow rhs\left(s(t)_3\right)$$

```
> plot([x,y,0..100],color=blue,thickness=2); # trajectoire de
  phases
```



Le mouvement semble converger vers un cycle limite stable. Par contre :

```
> init:=theta(0)=0.05,D(theta)(0)=0.03;
```

$$init := \theta(0) = 0.05,\ D(\theta)(0) = 0.03$$

```
> s:=dsolve({deq,init},theta(t),numeric):
> plot([x,y,0..57],color=blue,thickness=2);
```



Le bateau chavire. Représentons le naufrage en animation :

```
> bateau:=proc(theta)  # dessine un bateau incliné d'un angle
  theta
     local pts,rotation;
     pts:=[[0,0.5],[-1.2,0.5],[-1,-0.5],[0,-1],[1,-0.5],[1.2,
  0.5],[0,0.5],[0,3]];
     rotation:=P->[cos(theta)*P[1]-sin(theta)*P[2],sin(theta)
  *P[1]+cos(theta)*P[2]];
         # rotation d'angle theta
     plot(map(rotation,pts),thickness=10,color=brown,scaling=
  constrained)
  end:
> with(plots):
> dessins:=seq(bateau(x(k/5)),k=0..57*5):
> display(dessins,scaling=constrained,insequence=true);
```



Exercice 2 : équation KdV

```
> restart:
> pdeq:=diff(u(x,t),t)+6*u(x,t)*diff(u(x,t),x)+diff(u(x,t),
  x$3);
```

$$pdeq := \frac{\partial}{\partial t}\,u(x,t) + 6\,u(x,t)\left(\frac{\partial}{\partial x}\,u(x,t)\right) + \frac{\partial^3}{\partial x^3}\,u(x,t)$$

```
> with(PDEtools):
> sol:=pdsolve(pdeq,u(x,t));
```

$$sol := u(x,t) = -\frac{1}{6}\,\frac{\_C3 - 8\,\_C2^3}{\_C2} - 2\,\_C2^2\tanh(\_C1 + \_C2\,x + \_C3\,t)^2$$

```
> u:=unapply(subs(sol,u(x,t)),x,t);
```

$$u := (x,t) \to -\frac{1}{6}\,\frac{\_C3 - 8\,\_C2^3}{\_C2} - 2\,\_C2^2 \tanh(\_C1 + \_C2\,x + \_C3\,t)^2$$

```
> inits:=u(0,0)=1,D[1](u)(0,0)=0,D[1,1](u)(0,0)=-1;
```

$$inits := -\frac{1}{6}\,\frac{\_C3 - 8\,\_C2^3}{\_C2} - 2\,\_C2^2 \tanh(\_C1)^2 = 1,\ -4\,\_C2^3 \tanh(\_C1)\left(1 - \tanh(\_C1)^2\right) = 0,\ -4\,\_C2^4\left(1 - \tanh(\_C1)^2\right)^2 + 8\,\_C2^4 \tanh(\_C1)^2\left(1 - \tanh(\_C1)^2\right) = -1$$

```
> s:=solve({inits});
```

$$s := \{\_C2 = \mathrm{RootOf}\left(2\,\_Z^2 - 1, label = \_L2\right), \_C1 = 0, \_C3 = -2\,\mathrm{RootOf}\left(2\,\_Z^2 - 1, label = \_L2\right)\}, \{\_C2 = \mathrm{RootOf}\left(2\,\_Z^2 + 1, label = \_L3\right), \_C1 = 0, \_C3 = -10\,\mathrm{RootOf}\left(2\,\_Z^2 + 1, label = \_L3\right)\}$$

```
> ss:=allvalues(s[1]);
```

$$ss := \left\{\_C1 = 0, \_C2 = \frac{1}{2}\sqrt{2}, \_C3 = -\sqrt{2}\right\}, \left\{\_C1 = 0, \_C2 = -\frac{1}{2}\sqrt{2}, \_C3 = \sqrt{2}\right\}$$

```
> u1:=unapply(subs(ss[1],u(x,t)),x,t);
```

$$u1 := (x,t) \to 1 - \tanh\left(\frac{1}{2}\sqrt{2}\,x - \sqrt{2}\,t\right)^2$$

```
> with(plots):
> animate(plot,[u1(x,t),x=-10..10,thickness=2],t=0..5);
```




```
>
```